

الفصل الثانى


الاحصاء الوصفى وبرنامج الاكسيل


Descriptive Statistics with

Excel

الاحصاء الوصفي وبرنامج الاكسيل

Descriptive Statistics with Excel

يمكن استخدام برنامج الاكسيل في اجراء بعض التحليلات الاحصائية الاساسيه، حيث أن هذا البرنامج يتيح للمستخدم العديد من التحليلات الاحصائية الهامه. وذلك عن طريق ادخال البيانات ثم ادخال الداله الخاصه بالتحليل المطلوب في الخليه المطلوبه في ورقه العمل. واذا كانت هذه الداله المطلوبه غير متوافره في البرنامج فيوجد طريق اخر لعمل التحليل الاحصائي وهو استخدام **ساحر الدوال Function Wizard** والذي يقود المستخدم خطوه تلى الاخرى حتى يتم اجراء التحليل المطلوب. ولاستخدام ساحر الدوال يتم الضغط على مفتاح  والموجود في مسطرة الادوات أو استخدام **Insert / Function** من مسطرة القوائم. فتظهر شاشة ادراج داله **Insert Function** كما في الشكل التالي.



حيث تعرض هذه الشاشة ثلاثة اجزاء، هذه الاجزاء هي:

- **البحث عن داله Search for a function** ويمكن للمستخدم كتابة اسم الداله المطلوبه أو اسم التحليل المطلوب أو اى كلمه تعبر عن التحليل الاحصائي المطلوب ثم الضغط على مفتاح **أذهب Go** فيظهر اسم الداله أو الدوال التي تقوم بالتحليل المطلوب في الجزء الثالث من الشاشة والذي يسمى اختيار داله **Select a function**.

- أو اختيار مجموعة المعادلات Or select a category حيث يعرض هذا الجزء انواع الدوال المتاحة فى برنامج الاكسيل، وانواع هذه الدوال احدى عشرة نوع ، هذه الدوال يمكن عرضها فى الجدول الاتى:

اسم مجموعة المعادلات	الوظيفة
Financial	دوال رياضيات المال والاستثمار وعددها 90 داله.
Date & Time	دوال حساب الوقت والتواريخ وعددها 26 داله.
Math & Trig	دوال الرياضيات وحساب المثلثات والمصفوفات والمحددات وعددها 66 داله.
Statistical	دوال التحليل الاحصائى وعددها 80 داله.
Lookup & Reference	دوال تحويل النصوص الى ارقام والعكس وكذلك دوال المراجعه وعددها 18 داله.
Database	دوال قواعد البيانات وعمليات الفرز والفهرسه وعددها 12 داله.
Text	دوال تنسيق النصوص وعددها 24 داله.
Logical	الدوال المنطقيه وعددها 6 دوال.
Information	مجموعة دوال المعلومات والتي يمكن استخدامها فى تحديد الخلايا التى يمكن الرجوع اليها فى اجراء حسابات معينه وعددها 19 داله.
User Defined	مجموعة الدوال التى يستطيع الباحث اضافتها الى البرنامج حسب احتياج كل مستخدم.
Engineering	الدوال الهندسيه مثل تحويل الارقام العشريه الى ارقام ثنائيه أو ثمانيه وعددها 80 داله.

- الجزء الثالث من الشاشة يعرض امكانية اختيار الداله **Select a function** حيث يتم اختيار احد الدوال المعروضه ثم الضغط على مفتاح **Ok** ليبدأ سحر الدوال فى العمل خطوه تلى الاخرى حتى يتم اتمام التحليل.
- بهذا يكون قد تم تجهيز برنامج الاكسيل لعمل التحليلات الاحصائيه، وهذا ما سوف نتعرض له فى الاجزاء القادمه.

تعرضنا فى الفصل السابق لطرق تلخيص البيانات جدولياً باستخدام الجداول التكرارية، وبيانياً باستخدام المدرج التكرارى، والمضلع التكرارى والمنحنى المتجمع الصاعد والمنحنى المتجمع الهابط، وهذه الرسوم تعطى فكره جيده عن شكل وطبيعة البيانات. ولكن تبقى الحاجه الى تلخيص البيانات رقمياً. وهناك ثلاثة طرق للتلخيص الرقى للبيانات وهذه الطرق هى:

1. مقاييس النزعه المركزيه،

2. مقاييس التشتت،

3. ومقاييس شكل التوزيعات.


(1) مقاييس النزعه المركزيه

وهى تسمى بمقاييس الموضع أو المتوسطات **Averages** وهى تلخص البيانات عن طريق ايجاد قيمه متوسطه لها. ومن أكثر مقاييس النزعه المركزيه استخداماً ما يلى:

- | | |
|-------------------|--------------------|
| 1. Mode | 1. المنوال |
| 2. Median | 2. الوسيط |
| 3. Mean | 3. الوسط الحسابى |
| 4. Quartiles | 4. الربيعات |
| 5. Geometric Mean | 5. الوسط الهندسى |
| 6. Harmonic mean | 6. الوسط التوافيقى |
| 7. Percentiles | 7. المقياس المئىنى |
| 8. Trimmed mean | 8. الوسط المنسق |

(1-1) المنوال:

المنوال هو اكثر القيم حدوثاً أو تكراراً لمجموعه من البيانات. ويمكن استخدام المنوال للبيانات الاسميه، الترتيبيه، الفتريه والنسب. وإذا كان هناك توزيع له منوال واحد فهذا التوزيع يسمى احادى المنوال **Unimodal distribution**. أما اذا كان التوزيع له منوالان فيسمى توزيع ثنائى المنوال **Bimodal distribution**. ولقد لاحظ بعض الاحصائيين وجود اكثر من منوالان لبعض التوزيعات وفى هذه الحاله يسمى توزيع متعدد المنوال **Multimodal distribution**. (ويمكن ملاحظه ذلك فى الشكل التالى).



(2-1) الوسيط:

الوسيط هو القيمة التى تتوسط البيانات المرتبه تصاعدياً أو تنازلياً. ويمكن تطبيقه للبيانات الترتيبية **Ordinal**، الفترية **Interval** والنسب **Ratio** ولكن لايمكن تطبيقه للبيانات الاسمية **Nominal**. أن الوسيط لايتأثر بالقيم المتطرفه سواء كانت قيم متطرفه **Extreme values** صغيره أو قيم متطرفه كبيره.

خطوات حساب الوسيط

1. يتم ترتيب البيانات من الاصغر الى الاكبر (تصاعدياً) أو من الاكبر الى الاصغر (تنازلياً) ولكن الافضل الترتيب التصاعدى.
2. إذا كان عدد المشاهدات فردى فإن الوسيط هو المفرده التى تتوسط البيانات المرتبه.
3. إذا كان عدد المشاهدات زوجى فإن الوسيط هو المفردتان اللتين تتوسطا البيانات المرتبه.
4. يمكن ايجاد موقع الوسيط بأستخدام المعادله التاليه: $\frac{(n+1)}{2}$

(3-1) الوسط الحسابى:

الوسط الحسابى من اكثر مقاييس الموضع استخداماً. ويمكن حسابه عن طريق قسمة مجموع القيم على عددها. أن الوسط الحسابى للمجتمع هى مجموع القيم مقسومه على حجم المجتمع.

$$\mu = \frac{\sum X}{N}$$

حيث أن N تمثل حجم المجتمع.

ويمكن حساب الوسط الحسابى للعينه بأستخدام المعادله التاليه:


$$\bar{X} = \frac{\sum X}{n}$$

حيث أن n تمثل حجم العينه، وجدير بالذكر أن متوسط العينه يعتبر تقدير لمتوسط المجتمع.

(4-1) الربيعات

الربيعات تعتبر من مقاييس النزعه المركزيه والتي تقوم بتقسيم البيانات الى اربعة مجموعات فرعيه. الربيع الاول ويرمز له بالرمز Q_1 ونجد أن 25% من البيانات تكون اصغر من الربيع الاول.

الربيع الثانى أو الوسيط ويرمز له بالرمز Q_2 ونجد أن 50% من البيانات تكون اصغر من الربيع الثانى.
الربيع الثالث ويرمز له بالرمز Q_3 ونجد أن 75% من البيانات تكون اصغر من الربيع الثالث.



ونجد أن قيم الربيعات ليس من الضرورى أن تكون قيم فى فئة البيانات.

خطوات حساب الربيعات:

1. يتم ترتيب البيانات تصاعدياً.
2. يتم حساب ترتيب الربيع الاول $\frac{(n+1)}{4}$
3. يتم حساب ترتيب الربيع الثانى $\frac{(n+1)}{2}$
4. يتم حساب ترتيب الربيع الثالث $\frac{(n+1)*3}{4}$
5. يتم الحصول على قيم الربيعات من البيانات المُرْتَبَة.

(5-1) الوسط الهندسى:

الوسط الهندسى (GM) لمجموعه من الارقام هو الجذر النونى لحاصل ضرب n رقم. ويمكن حسابه بأستخدام المعادله التاليه:

$$GM = \sqrt[n]{(X_1)(X_2)(X_3).....(X_n)} = \sqrt[n]{\prod x_i} \quad \text{for } X_i > 0$$

ويمكن استخدام الوسط الهندسى لحساب متوسط النسب المئوية، الارقام القياسيه، ويشترط أن تكون جميع القيم اكبر من الصفر، وذلك لان وجود قيمه واحده على الاقل سالبه أو تساوى صفر سوف يؤدي الى عدم امكانية حساب الوسط الهندسى.

(6-1) الوسط التوافيقى

يمكن استخدام الوسط التوافيقى لحساب متوسط المعدلات الزمنية مثل متوسط معدل الانتاج كل ساعه، أو متوسط سرعة السياره كل ساعه، أو متوسط سرعة الطباعه بأستخدام ماكينات للطباعه حيث أن سرعة ماكينات الطباعه تحدد بورقه كل دقيقه، أو متوسط عدد المواليد كل يوم. والوسط التوافيقى هو ناتج قسمة عدد المفردات على مجموع مقلوب القيم والتي يمكن صياغته كما يلى:

$$\text{Harmonic mean (HM)} = \frac{n}{\sum_{i=1}^n \left(\frac{1}{x_i} \right)}$$

(7-1) المقياس المئىنى

هذا المقياس يقوم بتقسيم البيانات الى مائة جزء، وذلك بعد ترتيب البيانات ترتيباً تصاعدياً. ويمكن الحصول على سبيل المثال على المئين العاشر وهو المفردة العاشره فى البيانات المرتبه تصاعدياً. ويستخدم المقياس المئىنى للتعرف على المفردات المتطرفه فى البيانات. وذلك لانه عند ترتيب المفردات تصاعدياً فأن المفردات المتطرفه الصغيره تأتى فى اول الترتيب والمفردات المتطرفه الكبيره تأتى فى اخر البيانات. ويمكن استخدام المعادله التاليه لحساب ترتيب المئين المطلوب:

$$i = \frac{P}{100}(n)$$

حيث أن P هو الترتيب المئىنى المطلوب، n هو عدد المفردات.

(8-1) الوسط المنسق

1. يستخدم الوسط المنسق فى حالة احتواء البيانات على مفردات متطرفه، ويمكن حساب الوسط المنسق بأتباع الخطوات التاليه:
2. يتم ترتيب البيانات تصاعدياً.
3. يتم تحديد نسبة الحذف المطلوبه وتعتمد نسبة الحذف على خبرة الباحث بطبيعة البيانات وفى حالة عدم توافر الخبره فعلى الباحث ان يستخدم عدة نسب متدرجه حتى يحصل على قيمه ثابتة للمتوسط المنسق وفى هذه الحاله يكون قد تم تحديد نسبة المفردات المتطرفه فى البيانات.
4. يتم ازالة مفردات تعادل نصف نسبة الحذف من اول البيانات (المفردات الصغيره) والنصف الاخر من اخر البيانات (المفردات الكبيره).
5. يتم حساب الوسط الحسابى لباقي المفردات بعد الحذف.
6. يمكن استخدام المعادله التاليه لحساب الوسط المنسق:

$$\bar{X}(\alpha) = \frac{1}{n(1-2\alpha)} \left\{ (1-r)[X_{k+1} + X_{n-k}] + \sum_{i=k+2}^{n-k-1} X_i \right\}$$

Where

$$k = \text{Integer}(cn) = \text{Int}(cn) = [cn] \quad \text{and} \quad r = cn - k$$

(2) مقاييس التشتت

تقيس مقاييس التشتت درجة انتشار البيانات حول قيمه معينه غالباً ما تكون الوسط الحسابي. وتستخدم هذه المقاييس للتعرف على مدى انتظام البيانات، حيث أن البيانات المنتظمة بيانات لا تحتوى على مفردات متطرفه، اما البيانات الغير منتظمة فهي بيانات تحتوى على مفردات متطرفه. ونجد أن كلما زادت قيمة مقياس التشتت كلما دل ذلك على وجود مفردات متطرفه. ومن اكثر مقاييس التشتت استخداماً ما يلي:

1. Range 1. المدى
2. Interquartile Range 2. المدى الربيعي
3. Mean Absolute Deviation 3. متوسط الانحرافات المطلقة
4. Variance and Standard Deviation 4. التباين والانحراف المعياري
5. Standard error 5. الخطأ المعياري
6. Coefficient of Variation 6. معامل الاختلاف

(1-2) المدى

المدى هو الفرق بين القيمة الكبرى والقيمة الصغرى لفئه من البيانات.

(2-2) المدى الربيعي

المدى الربيعي هو المسافه بين الربع الاعلى (الربع الثالث) والربع الادنى (الربع الاول). هذه المسافه تحتوى على 50% من المشاهدات.

المدى الربيعي أو الانحراف الربيعي يمكن حسابه بالطريقه التاليه:

$$\text{Interquartile range} = \text{IQR} = Q_3 - Q_1$$

نصف المدى الربيعي يمكن حسابه بالمعادله التاليه:

$$\text{Semi - Interquartile range} = \text{SIR} = \frac{Q_3 - Q_1}{2}$$

(3-2) متوسط الانحرافات المطلقة:

هو الوسط الحسابي لمجموع الانحرافات المطلقة للقيم عن الوسط الحسابي.

$$MD = \frac{\sum |X - \bar{X}|}{n}$$

(4-2) التباين والانحراف المعياري:

أن التباين هو متوسط مربعات الانحرافات عن الوسط الحسابى. ويمكن حساب تباين المجتمع بالمعادله التاليه:

$$\sigma^2 = \frac{\sum (X - \mu)^2}{N}$$

الانحراف المعياري هو جذر التباين.

$$\sigma = \sqrt{\frac{\sum (X - \mu)^2}{N}}$$

يمكن حساب تباين العينه بأستخدام احد الصيغ التاليه:

$$S^2 = \frac{\sum (X - \bar{X})^2}{n-1} \text{ or } S^2 = \frac{\sum X^2 - n\bar{X}^2}{n-1}$$

يمكن حساب الانحراف المعياري للعينه بأستخدام احد الصيغ التاليه:

$$S = \sqrt{\frac{\sum (X - \bar{X})^2}{n-1}} \text{ or } S = \sqrt{\frac{\sum X^2 - n\bar{X}^2}{n-1}}$$

(5-2) معامل الاختلاف

معامل الاختلاف يسمى التشتت النسبى. ففى بعض الحالات يكون المطلوب المقارنه بين فئتين من البيانات مع اختلاف وحدات القياس لكل منهما فمثلاً يكون المطلوب المقارنه بين تشتت الاسعار لمجموعة سلع معينه، والكميات المباعه لنفس السلع. فنجد أن الاسعار مقاسه بالجنيه والكميات مقاسه بالكيلوجرام فلايمكن الاعتماد على مقاييس التشتت للمقارنه بين تشتت الاسعار وتشتت الكميات وهنا يجب استخدام مقياس نسبى وهو معامل الاختلاف. وهناك معادلتين يمكن استخدامهما لحساب معامل الاختلاف:

$$\text{Coefficient of variation} = CV = \frac{S}{\bar{X}} \times 100 \quad (1)$$

$$\text{Coefficient of variation} = CV = \frac{Q_3 - Q_1}{Q_3 + Q_1} \times 100 \quad (2)$$

المعادله الاولى تستخدم للبيانات التى لاتحتوى على مفردات متطرفه.

المعادله الثانيه تستخدم للبيانات التى تحتوى على مفردات متطرفه.

(3) مقاييس شكل التوزيع


أن التعرف على شكل توزيع البيانات تهم الباحث حيث أن توزيعات البيانات يمكن التفريقه بينهم بأستخدام طريقتين رئيسيتين هما:

1. Skewness
2. Kurtosis

1. مقياس الالتواء
2. مقياس التقاطح

(1-3) الالتواء


أن معامل الالتواء يقيس اتجاه ودرجة انتظام التوزيع. فإذا كانت قيمة هذا المقياس موجبه فهذا يعنى أن ذيل المنحنى الطويل يقع على يمين المنحنى، بينما القيمه السالبه تعنى أن ذيل المنحنى الطويل يقع على يسار المنحنى. أما فى حالة أن قيمة هذا المقياس تساوى صفر فهذا يعنى أن التوزيع معتدل وأن طرفى المنحنى متساويان، كما يتضح من الشكل التالى:



ولكن ماذا يعنى الالتواء السالب: إذا كان توزيع البيانات ملتوى ناحية اليسار فهذا يعنى ان القيم الكبيره تركيزها اكثر من القيم الصغيره.

ولكن ماذا يعنى الالتواء الموجب: إذا كان توزيع البيانات ملتوى ناحية اليمين فهذا يعنى ان القيم الصغيره تركيزها اكثر من القيم الكبيره.

الشكل التالى يوضح علاقه بين الوسط الحسابى والوسيط والمنوال فى حالة التواء التوزيع السالب والموجب وفى حالة التوزيع المنتظم أو الطبيعى.



من هذا الرسم نجد انه اذا كان:

الوسط الحسابى **اكبر من** الوسيط **اكبر من** المنوال فان التوزيع **موجب الالتواء**.

الوسط الحسابى **اصغر من** الوسيط **اصغر من** المنوال فان التوزيع **سالب الالتواء**.

الوسط الحسابى **يساوى** الوسيط **يساوى** المنوال فان **التوزيع معتدل**.

ويمكن حساب معامل الالتواء باستخدام المعادله التاليه:

$$Skewness\ coefficient = B_1 = \frac{n}{(n-1)(n-2)} \sum \left[\frac{x_i - \bar{x}}{s} \right]^3$$

(2-3) التفلطح

هذا المقياس يقيس كثافة ذيل التوزيع. أن التوزيع الطبيعى هو المرجع الذى يتم على اساسه مقارنة تفلطح اى توزيع للبيانات. فإذا كان معامل التفلطح يساوى 3 ومعامل الالتواء يساوى صفر فان التوزيع يكون طبيعى.

أن التوزيعات احادية المنوال والتي لها معامل تفلطح اكبر من 3 فان ذيل المنحنى يكون اكثر سمكاً من التوزيع الطبيعى. وهذه التوزيعات تميل الى ظهور قمه أعلى من قمة التوزيع الطبيعى وهذا يسمى منحنى **مدبب**

.Leptokurtic

أن التوزيعات احادية المنوال والتي لها معامل تفلطح اصغر من 3 فإن ذيل المنحنى يكون اقل سمكاً من التوزيع الطبيعي. وهذه التوزيعات تميل الى ظهور قمة اقل من قمة التوزيع الطبيعي وهذا يسمى منحنى **مفلطح**

.Platykurtic

بطريقه اخرى فإن قمة المنحنى للتوزيع يمكن تقسيمها الى ثلاثة انواع هي:

Leptokurtic

مدبب: القمه مرتفعه ومدببه


Mesokurtic

طبيعي: المنحنى يأخذ شكل التوزيع الطبيعي

Platykurtic

مفلطح: قمة المنحنى مفلطحه

كما يتضح من الشكل التالى:



وجدير بالذكر أن هذا المقياس يعطى تقديرات غيرمأمونه لتفلطح البيانات فى حالة احجام العينات الصغيره. ويمكن حساب معامل التفلطح بأستخدام المعادله التاليه:

Kurtosis coefficient =

$$B_2 = \left\{ \frac{n(n+1)}{(n-1)(n-2)(n-3)} \sum \left[\frac{x_i - \bar{x}}{s} \right]^4 \right\} - \frac{3(n-1)^2}{(n-2)(n-3)}$$

استخدام الاكسيل فى التحليل الوصفى

فى الجزء السابق تم شرح لمقاييس الاحصاء الوصفى نظرياً وتعرفنا على معنى كل مقياس واهميته وكيفية حسابه يدوياً، فى الجزء التالى سوف نتعرض لطريقة حساب هذه المقاييس بأستخدام برنامج الاكسيل. وجدير بالذكر أنه هناك طريقتان لعمل التحليل الاحصائى باستخدام الاكسيل هاتين الطريقتان هما:

1. استخدام تحليل البيانات من ادوات البرنامج **Tools / Data analysis**

2. استخدام الدوال الموجوده فى تصميم البرنامج **Built-in functions**

ونجد ان الفرق بين الطريقتان هو ان استخدام الطريقة الاولى سهل ولكن لايمكن عمل تحديث **Updating** للنتائج فى حالة تغير البيانات، ويجب على الباحث اعادة التحليل مره اخرى اذا حدث اى تغير أو تعديل للبيانات. أما الطريقة الثانيه فى تحتاج الى مهاره معينه يجب توافرها فى الباحث وتتميز بأن نتائجها متصله بالبيانات اتصال دائم **Linkage** بحيث أن اى تعديل أو تغير فى البيانات يؤثر فورياً على النتائج. وينصح بأستخدام الطريقة الاولى اذا كان الباحث متأكد من صحة البيانات وسوف يقوم بعمل التحليل الاحصائى مره واحده، اما اذا كان الباحث سوف يقوم بعمل التحليل الاحصائى اكثر من مره وأن بياناته دائمة التحديث فعليه استخدام الطريقة الثانيه. لشرح هاتين الطريقتين سوف يتم عرض المثال التالى لاستخدامه فى التطبيق العملى.

(1) مثال:

بأفتراض أن المبيعات الشهريه لاحد الشركات خلال العامين 2000، 2001 موضحه بالشكل التالى:

	A	B	C
1	Month	Year 2000	Year 2001
2	Jan	253	140
3	Feb	339	460
4	Mar	454	483
5	Apr	106	263
6	May	445	155
7	Jun	198	118
8	Jul	113	166
9	Aug	188	107
10	Sep	214	237
11	Oct	321	243
12	Nov	249	242
13	Dec	464	286

أولاً: تحليل البيانات من الادوات Tools / Data analysis

من المثال السابق المطلوب حساب مقاييس الاحصاء الوصفى لهذة البيانات.


يجب التأكد اولاً ان برنامج تحليل البيانات **Data Analysis** قد تم تحميله فى برنامج الاكسيل، ويمكن التأكد من ذلك اذا تم الضغط على مفتاح الادوات **Tools** فى مسطرة القوائم **Menu Bar** والموجوده فى اعلى الشاشة فستجد تحليل البيانات **Data Analysis** فى اخر قائمه، واذا لم تكن موجوده فعلى الباحث اتباع الخطوات التاليه:

اضافة التحليلات الاحصائيه والرياضيه الى برنامج الاكسيل:


ان برنامج الاكسيل يحتوى على بعض التحليلات الاحصائيه الاساسيه للباحثين، وللاستفاده منها واستخدامها يجب تجهيزها وذلك من خلال الخطوات الاتيه:

1. يتم الضغط على **Tools** من مسطرة القوائم ثم اختيار **Add-Ins** اضافه وادراج كما موضح بالشكل

التالى:



يظهر بعد ذلك الشاشة التاليه:



سنجد فى هذه الشاشة الاختيارات التالىة:

Analysis ToolPak: اداة التحليل الاحصائى


Analysis ToolPak VBA: استخدام لغة البيسك المرئى لتصميم ماكرو او برنامج يقوم بعمل معين.

Conditional Sum Wizard: ساحر ايجاد المجاميع الشرطيه.

Lookup Wizard: ساحر المقارنه

Solver Add-in: اضافة برنامج بحوث العمليات.

يتم اختيار هذه البرامج لاضافتها لبرنامج الاكسيل، واذا كانت مجموعة برامج **MS Office** تم تحميلها بطريقه كامله وذلك عند بداية تجهيز الحاسب فان برنامج الاكسيل سوف يستغرق عدة دقائق (تعتمد على سرعة اجهاز) وتظهر قائمة التحليل الاحصائى **Data Analysis** فى قائمة الادوات **Tools**. اما اذا كان تحميل برنامج **Office** غير كامل فان البرنامج سوف يطلب منك ادخال **CD** الخاص بمجموعة برامج **Office** حتى يستطيع تحميل البرامج المطلوبه، وفى كل الحالتين ستكون قائمة الادوات كما يلى:




لاحظ ما تم اضافته والذي لم يكن موجود من قبل.

لاحظ ان هذه العملية يتم عملها مره واحده ولا يحتاج الباحث الى اعاتتها فى كل مره يقوم فيها بعمل تحليل احصائى.


أولاً: استخدام تحليل البيانات Data Analysis

الان وقد تم تجهيز برنامج الاكسيل بالتحليل الاحصائى علينا ان نقوم بحساب مقاييس الاحصاء الوصفى لبيانات المثال السابق. لعمل التحليل الاحصائى على الباحث اتباع الخطوات التاليه:

- ❖ يتم الضغط على مفتاح الادوات **Tools** من مسطرة القوائم.
- ❖ يتم اختيار تحليل البيانات **Data Analysis** فنحصل على الشاشة التاليه:



- ❖ من هذه الشاشة يتم اختيار الاحصاء الوصفى **Descriptive Statistics** فتظهر الشاشة التاليه:



❖ هذه الشاشة مقسمه الى قسمين، القسم الاول للمدخلات **Input**، والقسم الثانى للمخرجات **Output**.

القسم الاول: قسم المدخلات Input:

مدى المدخلات Input Range فى هذا الجزء يتم تحديد مدى البيانات ومدى البيانات فى هذا المثال هو

B1:C13. لاحظ أن مدى البيانات يتضمن العنوان فى الخليه الاولى للعمود **B** والخلية الاولى للعمود **C**.

تجميع البيانات Grouped By: فى هذا الجزء لدينا اختيارين هما البيانات تم وضعها فى صورة اعمده

Columns أو تم وضعها فى صورة صفوف **Rows**، فى المثال الحالى تم وضع البيانات فى صورة

اعمده ولهذا تم اختيار الاعمده. وينصح ان تكون البيانات فى صورة اعمده وذلك تمشياً مع القواعد المتبعه

لتصميم قواعد البيانات وايضاً حتى يسهل تحويل البيانات من برنامج الاكسيل الى اى برنامج اخر.

العنوان فى الصف الاول Labels in first row إذا كانت الخليه الاولى فى كل عمود فيها عنوان العمود

يتم تنشيط هذا الاختيار، أما إذا مانت الخليه الاولى فى كل عمود أو فى بعض الاعمده لاتحتوى على العنوان

فلا يتم تنشيط هذا الاختيار.

القسم الثانى: قسم المخرجات Output:

مدى المخرجات Output Range: فى هذا الجزء يتم تحديد المكان المراد وضع النتائج فيه ويفضل

اختيار اى خليه فارغه بجوار البيانات.

عرض ورقة عمل جديده: New Worksheet Ply هذا الاختيار يقوم بعرض النتائج فى ورقة عمل جديده.

ملف جديد: New Workbook هذا الاختيار يقوم بعرض النتائج فى ملف جديد، ويفضل استخدام هذه الامكانيه فى حالة عدم الرغبه فى عرض البيانات على شخص ما مثل حالة عرض نتائج تحليل بيانات الايداعات والمسحوبات فى البنوك على اشخاص معينين ليس لهم الحق فى الاطلاع على بيانات العملاء.

ملخص الاحصاء Summary Statistic هذا الاختيار يعطى نتائج استخدام مقاييس الاحصاء الوصفى للبيانات، وفى حالة عدم استخدام هذه الامكانيه فأن البرنامج لن يقوم بحساب هذه المقاييس.

مستوى الثقة للوسط Confidence Level of mean هذا الاختيار يتيح للباحث تحديد مستوى الثقة المطلوب استخدامه لانشاء فترة ثقة للوسط الحسابى. ومن المعروف أن مستويات الثقة المستخدمه فى البحوث الاحصائيه هى 90%، 95%، 99%، 99.7%. فإذا كانت البيانات تعبر عن استطلاع الرأى فى سلعه معينه أو ما شابه ذلك فيفضل استخدام معامل ثقة 90%، اما البحوث الاجتماعيه فيفضل استخدام معامل ثقة 95%.

المفرده الكبرى رقم Kth Largest تحديد اكبر مفرده والتي يكون رقمها هو **K**، امام هذا الاختيار صندوق يتم كتابة رقم المفرده الكبرى المطلوب، فإذا تم كتابة الرقم 1 فهذا يعنى تحديد اكبر مفرده، وأذا تم كتابة الرقم 2 فهذا يعنى تحديد ثانى اكبر مفرده اى المفرده التى تسبق المفرده الكبرى وهكذا.

المفرده الصغرى رقم Kth Smallest تحديد اصغر مفرده والتي يكون رقمها هو **K**، امام هذا الاختيار صندوق يتم كتابة رقم المفرده الصغرى المطلوب، فإذا تم كتابة الرقم 1 فهذا يعنى تحديد اصغر مفرده، وأذا تم كتابة الرقم 2 فهذا يعنى تحديد ثانى اصغر مفرده اى المفرده التى تسبق المفرده الصغرى وهكذا.

يتم الضغط على مفتاح الموافقه **Ok** لنحصل على النتائج التاليه:

	E	F	G	H
1	Year 2000		Year 2001	
2				
3	Mean	278.67	Mean	241.67
4	Standard Error	36.50	Standard Error	35.39
5	Median	251	Median	239.5
6	Mode	#N/A	Mode	#N/A
7	Standard Deviation	126.45	Standard Deviation	122.61
8	Sample Variance	15990.61	Sample Variance	15032.42
9	Kurtosis	-1.15	Kurtosis	0.45
10	Skewness	0.29	Skewness	1.07
11	Range	358	Range	376
12	Minimum	106	Minimum	107
13	Maximum	464	Maximum	483
14	Sum	3344	Sum	2900
15	Count	12	Count	12
16	Largest(1)	464	Largest(1)	483
17	Smallest(1)	106	Smallest(1)	107
18	Confidence Level(95.0%)	80.35	Confidence Level(95.0%)	77.90

التعليق على النتائج:

العمودان E، F يحتويان على مقاييس الاحصاء الوصفى لبيانات عام 2000.

العمودان G، H يحتويان على مقاييس الاحصاء الوصفى لبيانات عام 2001.

بيانات عام 2000:

نجد أن متوسط المبيعات بلغ 278.67 جنيه، بينما الوسيط بلغ 251، ونجد أنه لا يوجد منوال وذلك لعدم وجود أى مفرد مكرره فى هذه السنة. ونجد أن الانحراف المعياري بلغ 126.45، والتباين بلغ 15990.61، والخطأ المعياري **Standard Error** بلغ 36.50. والخطأ المعياري هو ناتج قسمة الانحراف المعياري على الجذر التربيعي لحجم العينة وهو جزء من معادلة فترة الثقة للوسط الحسابي. كما وجد أن معامل الالتواء يساوى 0.29 وهذا يعنى ان هذا التوزيع ملتوى التواء بسيط ناحية اليمين، ونجد أن معامل التقلطح بلغ -1.15 وهذا المعامل اصغر من 3 مما يعنى أن التوزيع مفلطح. ووجد أن القيمة الصغرى 106 والقيمة الكبرى 464 مما يؤدي الى ان المدى يصبح 358. أما فترة الثقة للوسط الحسابي **Confidence Level (95%)** فهذا هو الجزء الذى يتم

اضافته الى الوسط الحسابى لنحصل على الحد الاعلى ويتم طرحه من الوسط الحسابى لنحصل على الحد الادنى،
وعليه فإن حدى فترة الثقة هما:

$$\text{الحد الاعلى: } 359.02 = 80.35 + 278.67$$


$$\text{الحد الادنى: } 198.32 = 80.35 - 278.67$$

أى أن المبيعات فى عام 2000 تتراوح بين 198.32 و 359.02 باحتمال 95%.

ثانياً: استخدام الدوال الموجودة فى تصميم البرنامج Built-in functions

أن استخدام الدوال الموجودة فى برنامج الاكسيل يتيح للباحث فرصة تعديل البيانات والحصول على النتائج مباشرةً دون الحاجة الى اجراء التحليل مره اخرى. ونجد أن كل مقياس من المقاييس السابق ذكرها له الداله الخاصه به، ويمكن الحصول على الداله المطلوبه بالضغط على مفتاح ادراج داله **Insert Function** والموجود فى مسطرة

الادوات ويأخذ الشكل  فتظهر لنا الشاشة التاليه:



فى الجزء الاول من الشاشة **Search for a function** يتم كتابة اسم الداله اذا كان الاسم معروف لدى المستخدم، أو يتم كتابة كلمات تدل على الداله ثم الضغط على مفتاح **Go**. فيظهر اسم الداله فى الجزء الخاص باختيار داله **select a function** ثم يتم الضغط على مفتاح الموافقه **Ok**.


أو يتم اختيار القسم الذى يتتنمى اليه الداله من الجزء الخاص باختيار القسم **Or select a category**.

مقاييس النزعه المركزيه

فى هذا الجزء سوف يتم التعرض لمقاييس النزعه المركزيه بأستخدام الدوال المتوافره فى برنامج الاكسيل.

(1) دالة المنوال Mode Function:

- ◇ أن هذه الداله تقوم بحساب القيمه الاكثر شيوعاً أو تكراراً فى البيانات.
- ◇ هذه الداله تقوم بحساب منوال الخلايا المشغوله فقط وتتجاهل الخلايا الفارغه أو الخلايا التى تحتوى على نصوص والتى تقع فى المدى المحدد للداله.
- ◇ فى حالة عدم تكرار أى قيمه اكثر من مره واحده فإن الداله تعطى الاجابه "لايوجد" أو Null وتكتب كما يلي #N/A.
- ◇ فى حالة تكرار عدة قيم بنفس عدد المرات، فمثلاً القيمه 5 كان تكرار ظهورها 3 مرات، القيمه 6 كان تكرار ظهورها 3 مرات، القيمه 7 كان تكرار ظهورها 3 مرات. فإن دالة المنوال تعود باصغر قيمه مكرره وبهذا يكون المنوال هو القيمه 5.
- ◇ هذه الداله يمكنها حساب المنوال لعدد 30 عمود من القيم كحد اقصى.
- ◇ شاشة دالة المنوال تأخذ الشكل التالى:



(2) مثال:

	A	B	
1	1		
2	2	=MODE(A1:A5)	#N/A
3	3	=MODE(A6:A10)	6
4	4	=MODE(A6:A13)	6
5	5	=MODE(A6:A20)	6
6	6		
7	6		
8	7		
9	7		
10	10		
11	TRUE		
12	FALSE		
13	Medhat		

في هذا المثال سنجد ما يلي:

- المعادله الاولى **=MODE(A1:A5)** تقوم بحساب المنوال للقيم الموجوده فى المدى **A1:A5** وحيث انه لا توجد أى قيمه مكرره فأن الاجابه سوف تكون لا يوجد منوال **#N/A**.
- المعادله الثانيه **=MODE(A6:A10)** تقوم بحساب المنوال للقيم الموجوده فى المدى **A6:A10**، فى هذا المدى نجد ان هناك قيمتان متكررتان بنفس عدد مرات التكرار وهما المفرده التى قيمتها 6 والمفرده التى قيمتها 7، وحيث أن هذه الداله تأخذ اصغر القيم المكرره بنفس عدد مرات التكرار فيكون المنوال هو المفرده التى قيمتها 6.
- المعادله الثالثه **=MODE(A6:A13)** تقوم بحساب المنوال للقيم الموجوده فى المدى **A6:A13**، فى هذا المدى توجد خلايا تحتوى على نصوص **Text**، فنجد أن دالة المنوال تتجاهل الخلايا الغير رقميه وبهذا فأن المنوال مازال القيمه 6 وهو منوال الخلايا الرقميه المتوافره فى هذا المدى.
- المعادله الرابعه **=MODE(A6:A20)** تقوم بحساب المنوال للقيم الموجوده فى المدى **A6:A20**، فى هذا المدى توجد خلايا تحتوى على نصوص **Text**، وخلايا فارغه، فنجد أن دالة المنوال تتجاهل الخلايا الغير رقميه والخلايا الفارغه، وبهذا فأن المنوال مازال القيمه 6 وهو منوال الخلايا الرقميه المتوافره فى هذا المدى.

(2) دالة الوسيط Median Function:


- ◇ أن هذه الداله تقوم بحساب وسيط القيم.
- ◇ فإذا كان عدد المفردات فردى فإن الوسيط هو المفرده التى تتوسط البيانات بعد ترتيبها.
- ◇ إذا كان عدد المفردات زوجى فإن الوسيط هو الوسط الحسابى للمفردتين الاتين تتوسطا المفردات بعد ترتيبها.
- ◇ هذه الداله تقوم بحساب وسيط الخلايا المشغوله فقط وتتجاهل الخلايا الفارغه إذا كان هناك خلايا فارغه فى المدى المستخدم فى الداله. فمثلاً إذا كان هناك خمس قيم المطلوب حساب الوسيط لهم وتم تحديد المدى المستخدم للداله بالمدى **A1:A100** فإن الداله تقوم بحساب وسيط الخلايا المشغوله فقط.

(3) مثال:

بفرض انه لدينا المفردات التاليه وهى الارقام بالتسلسل من واحد الى خمسه وكان المطلوب حساب الوسيط لها. أن وسيط الارقام من واحد الى خمسه هو الرقم 3.

بفرض ادخال البيانات فى المدى **A1:A5** وتم استدعاء دالة الوسيط كما يلى:

Insert Function / Statistical Functions / Median فسوف نحصل على الشاشة التاليه:



يتم تحديد مدى البيانات عند **Number1** وجدير بالذكر أن هذه الداله يمكنها حساب الوسيط لعدد 30 عمود من القيم. ونجد فى اسفل الشاشة أن قيمة الوسيط **Formula result = 3** بالضغط على مفتاح الموافقه نحصل على النتيجة.

(3) دالة الوسط الحسابى Average Function:

- ◇ أن دالة الوسط الحسابى تقوم بحساب ناتج قسمة مجموع القيم على عددها لتعطى الوسط الحسابى.


◇ أن هذه الدالة تقوم بحساب الوسط الحسابي لمجموعة الخلايا المشغولة فقط، أي انه اذا وجدت خليه فارغه
فإن هذه الداله تقوم باهمالها. فعلى سبيل المثال أن الوسط الحسابي للارقام من 1 الى 5 يساوى 3، فإذا تم
ادخال هذه البيانات فى المدى **A1:A10** وهذا المدى يحتوى على عدد من الخلايا اكبر من عدد المفردات
وتم استخدام هذه الداله لهذا المدى فأنها تعطى الوسط الحسابى يساوى 3. أو بأسلوب اخر أن هذه الداله تقوم
بحساب مجموع الخلايا المشغوله على عدد الخلايا المشغوله.

◇ ونجد أن هناك نوعين من دوال الوسط الحسابى هما:

◇ **AVERAGE** هذه الداله تتعامل مع القيم الرقمية فقط ولا تتعامل مع النصوص.

◇ **AVERAGEA** هذه الداله تتعامل مع القيم الرقمية والنصوص فنجد انها تعطى الخليه التى تحتوى على
اللفظ **FALSE** قيمه صفر، والخليه التى تحتوى على اللفظ **TRUE** قيمه واحد.

لاستدعاء دالة الوسط الحسابى يتم الضغط على مفتاح ادراج داله **Insert Function** ثم اختيار الدوال الاحصائيه
Statistical Functions ثم اختيار دالة الوسط الحسابى للقيم فقط **Average** أو اختيار دالة الوسط الحسابى
للقيم والنصوص **Averagea** حسب طبيعة البيانات. فنحصل على الشاشة التاليه:



و جدير بالذكر أن اقصى عدد من الاعمده يمكن استخدامه لهذه الداله هو 30 عمود من القيم وعدد القيم بكل عمود
غير محدد، ويقصد بـ **Number1** عمود القيم الاول وهكذا.

(4) مثال:

دعنا نوضح ذلك بالمثال التالى:

	A	B	C	
1	1		=AVERAGE(A1:A5)	3
2	2		=AVERAGE(A1:A10)	3
3	3		=AVERAGEA(A1:A8)	2
4	4		=AVERAGEA(A1:A6)	2.50
5	5			
6	Medhat			
7	FALSE			
8	TRUE			

- المعادله الاولى =AVERAGE(A1:A5) تأخذ مدى البيانات التى تم ادخالها فى الخمس خلايا الاولى من العمود A وتعطى ناتج حساب مجموع هذه القيم على عددها أو تعطى الوسط الحسابى لهذه القيم.
- المعادله الثانيه =AVERAGE(A1:A10) تأخذ مدى البيانات من الخليه A1 الى الخليه A10 وهذا المدى يحتوى على خلايا عددها اكبر من عدد المفردات المراد حساب وسطها، فنجد أن الخلايا A6:A8 بها نصوص Text والخلايا A9:A10 خلايا فارغه، ولكن الناتج هو الوسط الحسابى للقيم الخمسه التى تم ادخالهم، وهذا يعنى أن هذه الداله قد تجاهلت النصوص الموجوده وايضاً تجاهلت الخلايا الفارغه، أى انها تقوم بحساب مجموع الخلايا المشغوله ويتم القسمة على عدد الخلايا المشغوله.
- المعادله الثالثه =AVERAGEA(A1:A8) هى معادلة حساب الوسط الحسابى للقيم الرقمية والنصوص، فنجد أن المدى المستخدم لهذه الداله يحتوى فى الخمس خلايا الاولى على قيم رقميه هى الارقام من 1 الى 5، وفى الثلاث خلايا الاخيريه توجد نصوص هى Medhat، False، True. أن هذه الداله تعطى القيمه واحد للفظ True، والقيمه صفر للفظ False أو اى لفظ اخر غير معرف للداله، وعلى هذا فإن الوسط الحسابى الذى تم حسابه هو:

$$\bar{X} = \frac{1+2+3+4+5+0+0+1}{8} = \frac{16}{8} = 2$$


- المعادله الرابعه =AVERAGEA(A1:A6) هى معادلة حساب الوسط الحسابى للقيم الرقمية والنصوص، فنجد أن المدى المستخدم لهذه الداله يحتوى فى الخمس خلايا الاولى على

قيم رقميه هي الارقام من 1 الى 5، وفي الخلايا الاخيريه يوجد نص هو Medhat. أن هذه الداله تعطى القيمه صفر للفظ False أو اى لفظ اخر غير معرف للداله، وعلى هذا فإن الوسط الحسابى الذى تم حسابه هو:

$$\bar{X} = \frac{1+2+3+4+5+0}{6} = \frac{15}{6} = 2.5$$

(4) دالة الربعيات Quartile Function

- ◇ هذه الداله تقوم بحساب الربع الاول، الربع الثانى (الوسيط)، الربع الثالث لمجموعه من البيانات.
 - ◇ هذه الداله تتجاهل الخلايا الفارغه والخلايا النصيه.
 - ◇ فى حالة تحديد رقم الربع المطلوب **Quart** برقم كسرى فإن البرنامج يقوم بعملية قطع **Truncate** لهذا الرقم.
 - ◇ إذا كان رقم الربع **Quart** اصغر من صفر أو اكبر من 4 فإن الداله تعطى خطأ رقمى **#NUM!**.
 - ◇ هذه الداله تعطى القيمة الصغرى **Min** للبيانات، الوسيط **Median**، والقيمه الكبرى **Max** فى حالة أن رقم الربع **Quart** يساوى صفر، 2، 4 على الترتيب.
- شكل شاشة هذه الداله كما يلى:



فى هذه الشاشة نجد أن **Array** تعنى مدى البيانات المستخدمه، ونجد أن **Quart** تعنى رقم الربع المطلوب حسابه ويجب ادخال رقم صحيح فى هذه الخانه فالرقم (1) يعنى الربع الاول، الرقم (2) يعنى الربع الثانى أو الوسيط، الرقم (3) يعنى الربع الثالث. مثال:

(5) مثال:

افترض أن لدينا البيانات التاليه:

42	68	20	75	25	30	52	46	55
----	----	----	----	----	----	----	----	----

وكان المطلوب حساب الرُّبِيعات الثلاثه وحساب المدى الرُّبِيعى ونصف المدى الرُّبِيعى لهذة البيانات.

الحل:

يتم ادخال البيانات فى ورقة عمل وكتابة المعادلات بالشكل التالى:

	A	B	
1	42		
2	68	=QUARTILE(A1:A9;1)	25
3	20	=QUARTILE(A1:A9;2)	42
4	75	=QUARTILE(A1:A9;3)	52
5	25	=B4-B2	25
6	30	=(B4-B2)/2	12.5
7	52		
8	46	=QUARTILE(A1:A15;1)	25
9	55	=QUARTILE(A1:A15;2)	42
10		=QUARTILE(A1:A15;3)	52
11	TRUE		
12	FALSE		
13	Medhat		

من هذا الجدول يتضح ما يلى:

- مدى البيانات هو **A1:A9** ويلاحظ أن عدد المفردات فردى.
- المعادله الاولى **=QUARTILE(A1:A9;1)** تقوم بحساب الرُّبِيع الاول، حيث أن رقم الرُّبِيع فى المعادله يساوى 1.
- المعادله الثانيه **=QUARTILE(A1:A9;2)** تقوم بحساب الرُّبِيع الثانى أو الوسيط، حيث أن رقم الرُّبِيع فى المعادله يساوى 2.
- المعادله الثالثه **=QUARTILE(A1:A9;3)** تقوم بحساب الرُّبِيع الثانى ، حيث أن رقم الرُّبِيع فى المعادله يساوى 3.
- المعادله الرابعه **=B4-B2** تقوم بحساب المدى الرُّبِيعى **IQR** وهى معادله اجرائيه اى تعتمد على مكان الرُّبِيع الاول ومكان الرُّبِيع الثالث فى ورقة العمل.
- المعادله الخامسه **=(B4-B2)/2** تقوم بحساب نصف المدى الرُّبِيعى **SIR** وهى معادله اجرائيه اى تعتمد على مكان الرُّبِيع الاول ومكان الرُّبِيع الثالث فى ورقة العمل.

- المعادلات من السادس الى الثامن تقوم بحساب الرُّبِيعات الثلاثة بالترتيب وبأستخدام مدى البيانات **A1:A15** وهذا المدى يحتوى على خلايا نصيه وخلايا فارغه، ولكن هذه الداله تتعامل مع البيانات الرقمية فقط وتقوم بأهمال الخلايا نصيه والخلايا فارغه. لاحظ الفرق بين الاجابات لهذه المعادلات والمعادلات الثلاثة الاولى.

(6) مثال:

افترض أن لدينا البيانات التاليه:

42	68	20	75	25	30	52	46	55	25
----	----	----	----	----	----	----	----	----	----

وكان المطلوب حساب الرُّبِيعات الثلاثة وحساب المدى الرُّبِيعى ونصف المدى الرُّبِيعى لهذه البيانات.

الحل:

يتم ادخال البيانات فى ورقة عمل وكتابة المعادلات بالشكل التالي:

	A	B	
1	42		
2	68	=QUARTILE(A1:A10;1)	26.25
3	20	=QUARTILE(A1:A10;2)	44
4	75	=QUARTILE(A1:A10;3)	54.25
5	25	=B4-B2	28
6	30	=(B4-B2)/2	14
7	52		
8	46		
9	55		
10	25		


من هذا الجدول يتضح ما يلى:

- مدى البيانات هو **A1:A10** ويلاحظ أن عدد المفردات زوجى.
- المعادله الاولى تقوم بحساب الرُّبِيع الاول، حيث أن رقم الرُّبِيع فى المعادله يساوى 1.
- المعادله الثانيه تقوم بحساب الرُّبِيع الثانى أو الوسيط، حيث أن رقم الرُّبِيع فى المعادله يساوى 2.
- المعادله الثالثه تقوم بحساب الرُّبِيع الثانى ، حيث أن رقم الرُّبِيع فى المعادله يساوى 3.
- المعادله الرابعه **=B4-B2** تقوم بحساب المدى الرُّبِيعى **IQR** وهى معادله اجرائيه اى تعتمد على مكان الرُّبِيع الاول ومكان الرُّبِيع الثالث فى ورقة العمل.

المعادله الخامسه $= (B4-B2)/2$ تقوم بحساب نصف المدى الربيعى SIR وهى معادله اجرائيه
 اى تعتمد على مكان الربيع الاول ومكان الربيع الثالث فى ورقة العمل.

(5) دالة الوسط الهندسى Geometric Mean Function

- ◇ هذه الداله تقوم بحساب الوسط الهندسى للخلايا المشغوله فقط وتتجاهل الخلايا النصيه والخلايا الفارغه.
- ◇ هذه الداله تعطى خطأ رقمى #NUM! فى حالة وجود مفرد واحد على الاقل سالبه او قيمتها صفر.
- ◇ هذه الداله يمكنها حساب الوسط الهندسى لعدد 30 عمود من القيم كحد اقصى.
- ◇ شاشة دالة الوسط الهندسى تأخذ الشكل التالى:



(7) مثال:

افترض أن لدينا البيانات التاليه والتي تم ادخالها فى ورقة عمل:


	A	B	
1	1	=GEOMEAN(A1:A3)	1.8171
2	2	=GEOMEAN(A1:A4)	#NUM!
3	3	=GEOMEAN(A5:A7)	#NUM!
4	-6	=GEOMEAN(A6:A11)	1.8171
5	0		
6	1		
7	2		
8	TRUE		
9	FALSE		
10	Medhat		
11	3		

من هذا الجدول يتضح ما يلي:

- المعادلة الاولى **=GEOMEAN(A1:A3)** تقوم بحساب الوسط الهندسى للقيم الموجوده فى المدى **A1:A3**، ويلاحظ أن جميع القيم موجبه ولا توجد قيم سالبه او تساوى صفر.
- المعادلة الثانيه **=GEOMEAN(A1:A4)** تقوم بحساب الوسط الهندسى للقيم الموجوده فى المدى **A1:A4**، ويلاحظ وجود احد القيم سالبه وهذا يؤدي الى ظهور خطأ رقمى **#NUM!**.
- المعادلة الثالثه **=GEOMEAN(A5:A7)** تقوم بحساب الوسط الهندسى للقيم الموجوده فى المدى **A5:A7**، ويلاحظ وجود احد القيم تساوى صفر وهذا يؤدي الى ظهور خطأ رقمى **#NUM!**.
- المعادلة الرابعه **=GEOMEAN(A6:A11)** تقوم بحساب الوسط الهندسى للقيم الموجوده فى المدى **A6:A11**، فى هذا المدى خلايا رقميه، وخلايا نصيه، ولكن هذه الداله تقوم بأنتقاء الخلايا الرقميه وتتجاهل الخلايا النصيه لحساب الوسط الهندسى، ارجع الى المعادلة الاولى وقارن بينها وبين هذه المعادله.

(6) دالة الوسط التوافيقى **Harmonic Mean Function**:

- ◇ هذه الداله تقوم بحساب الوسط التوافيقى للخلايا المشغوله فقط وتتجاهل الخلايا النصيه والخلايا الفارغه.
- ◇ هذه الداله تعطى خطأ رقمى **#NUM!** فى حالة وجود مفرده واحده على الاقل سالبه أو قيمتها صفر.
- ◇ هذه الداله يمكنها حساب الوسط التوافيقى لعدد 30 عمود من القيم كحد اقصى.
- ◇ الوسط التوافيقى اصغر من الوسط الهندسى اصغر من الوسط الحسابى.
- ◇ شاشة دالة الوسط التوافيقى تأخذ الشكل التالى:



(8) مثال:

افترض أن لدينا البيانات التاليه والتي تم ادخالها فى ورقة عمل:

	A	B	
1	1		
2	2	=HARMEAN(A1:A5)	2.189781
3	3	=HARMEAN(A6:A10)	#NUM!
4	4	=HARMEAN(A8:A13)	1.636364
5	5	=HARMEAN(A8:A15)	2.189781
6	-6	=HARMEAN(A14:A16)	#NUM!
7	-6		
8	1		
9	2		
10	3		
11	TRUE		
12	FALSE		
13	Medhat		
14	4		
15	5		
16	0		

من هذا الجدول نجد أن:

- المعادله الاولى **=HARMEAN(A1:A5)** تقوم بحساب الوسط التوافيقى للبيانات الموجوده فى المدى **A1:A5**، ويلاحظ أن جميع القيم الكبر من الصفر.
- المعادله الثانيه **=HARMEAN(A6:A10)** تقوم بحساب الوسط التوافيقى للبيانات الموجوده فى المدى **A6:A10**، فى هذا المدى توجد قيم سالبه وهذا يودى الى ظهور خطأ رقمى كنتيجه لاستخدام هذه الداله.
- المعادله الثالثه **=HARMEAN(A8:A13)** تقوم بحساب الوسط التوافيقى للبيانات الموجوده فى المدى **A8:A13**، هذا المدى يحتوى على خلايا نصيه **Text** ولكن دالة الوسط التوافيقى تتجاهل الخلايا النصيه وتقوم بحساب الوسط التوافيقى للخلايا الرقميه فقط.
- المعادله الرابعه **=HARMEAN(A8:A15)** تقوم بحساب الوسط التوافيقى للبيانات الموجوده فى المدى **A8:A15**، حيث توجد فى هذا المدى خلايا رقميه ثم خلايا نصيه ثم خلايا رقميه مره اخرى، ولكن هذه الداله تتلقى الخلايا الرقميه حتى لو لم تكن متتابعه وتتجاهل الخلايا النصيه.


- المعادله الخامسه =HARMEAN(A14:A16) تقوم بحساب الوسط التوافيقى للبيانات الموجوده فى المدى A14:A16، فى هذا المدى توجد خليه واحده على الاقل تساوى صفر، وبالتالي فأن دالة الوسط التوافيقى تعطى خطأ رقمى #NUM!.

(7) المقياس المئىنى Percentiles

- هذه الداله تقوم بحساب المئىن الذى له ترتيب معين لمجموعه بيانات مُرتبه تصاعدياً، فمثلاً يمكن استخدامه لتحديد الطلاب الحاصلين على درجات أعلى من المئىن التسعون.
- هذه الداله تقوم بحساب المئىن لعدد 8191 مفرده كحد اقصى، وفى حالة زياده عدد المفردات عن هذا الرقم فأن الداله تعطى خطأ رقمى #NUM!.
- إذا كان ترتيب المئىن قيمه غير رقميه فأن الداله تعطى خطأ كى #VALUE!.
- إذا كان ترتيب المئىن اصغر من الصفر أو اكبر من الواحد الصحيح فأن الداله تعطى خطأ رقمى #NUM!.

هذه الداله تتجاهل الخلايا الفارغه والخلايا النصيه.

ونجد أن شاشه هذه الداله تأخذ الشكل التالى:



- حيث أن **Array** تعنى القيم أو المفردات المطلوب حساب المئىن لها.
- كما أن **K** هى ترتيب المئىن المطلوب وهذه القيمه يجب أن تكون اكبر من الصفر واصغر من أو تساوى 1.

(9) مثال:


بفرض أن لدينا القيم التاليه وكان المطلوب حساب المئين الخامس والعشرون، والمئين الخمسون، والمئين الخامس والسبعون.

	A	B	
1	Data		
2	1	=PERCENTILE(A2:A5;0.25)	1.75
3	2	=PERCENTILE(A2:A5;0.5)	2.5
4	3	=PERCENTILE(A2:A5;0.75)	3.25
5	4		

- المعادله الاولى =PERCENTILE(A2:A5;0.25) تقوم بحساب المئين الخامس والعشرون وقيمته تساوى قيمة الربع الاول.
- المعادله الثانيه =PERCENTILE(A2:A5;0.5) تقوم بحساب المئين الخمسون وقيمته تساوى قيمة الربع الثانى أو الوسيط.
- المعادله الثالثه =PERCENTILE(A2:A5;0.75) تقوم بحساب المئين الخامس والسبعون وقيمته تساوى قيمة الربع الثالث.

(8) دالة الوسط المنسق Trimmed Mean Function

- ◇ هذه الداله تقوم بحساب الوسط المنسق لمجموعة الخلايا الرقيه فقط، وتتجاهل الخلايا النصيه والخلايا الفارغه.
 - ◇ هذه الداله تتطلب مجموعه من البيانات وتحديد نسبة الحذف المستخدمه، وفى حالة أن نسبة الحذف صفر فهذا يعنى حساب الوسط الحسابى للقيم المستخدمه.
 - ◇ فى حالة أن نسبة الحذف اصغر من صفر أو اكبر من واحد فإن الداله تعطى خطأ رقمى **!NUM#**.
 - ◇ هذه الداله تقوم بتقريب عدد المفردات الواجب حذفها الى اقرب رقم من مضاعفات الرقم 2 ويكون اصغر منه، فمثلاً إذا كان عدد المفردات 30 مفرده وكانت نسبة الحذف 10% فهذا يعنى حذف ثلاث مفردات، فإن هذه الداله تقوم بتقريب عدد المفردات الواجب حذفها الى 2 ويتم حذف مفرده من الامام ومفرده من الخلف.
- شاشة هذه الداله تأخذ الشكل التالي:



حيث أن **Array** تعنى مدى البيانات، بينما **Percent** تعنى نسبة الحذف المستخدم.

(10) مثال:

افتراض أن لدينا البيانات التالية، وكان المطلوب تحديد نسبة المفردات المتطرفه فيها:

20	211	306	339	373	402	466	532	560	1992
25	214	307	345	380	404	469	534	561	1994
25	252	325	359	397	431	501	551	583	2998
30	235	311	346	385	408	480	536	564	2995
42	247	317	353	389	416	495	542	578	4996
46	244	313	347	386	412	488	540	570	3996
52	305	334	367	401	437	519	553	594	2999
55	249	325	355	391	419	501	548	582	5996
68	305	336	370	401	446	519	558	598	3999
75	281	328	364	400	436	514	553	588	3999

الحل:

- أن عدد هذه المفردات 100 مفرده.
- لحساب الوسط المنسق يتم تلوّتيب البيانات ترتيباً تصاعدياً.
- يتم تحديد نسبة الحذف وهى نسبه مئوية.
- يتم حساب عدد المفردات الواجب حذفها، وهو حاصل ضرب نسبة الحذف فى عدد المفردات.
- يتم حذف نصف عدد المفردات الواجب حذفها من القيم الصغرى، ويتم حذف النصف الاخر من المفردات الكبرى وذلك من المفردات التى تم تلوّتيبها تصاعدياً.

- يتم حساب الوسط الحسابى للمفردات الباقية، وهو مجموع قيم المفردات الباقية مقسوماً على عدد المفردات الباقية.
- يتم زيادة نسبة الحذف وحساب الوسط المنسق.
- يتم تكرار الخطوه السابقه حتى نحصل على قيمتان متساويتان للوسط المنسق مع اختلاف نسب الحذف، وبهذا يتم تحديد نسبة الحذف أو نسبه المفردات المتطرفه وهى النسبه الاقل فى النسبتين الاخيرتين.
- بأفترض أن البيانات تم ادخالها فى المدى **A1:J10**، وتم كتابة المعادلات التاليه:

=AVERAGE(A1:J10)	699.83
=TRIMMEAN(A1:J10;0)	699.83
=TRIMMEAN(A1:J10;0.05)	614.02
=TRIMMEAN(A1:J10;0.10)	520.61
=TRIMMEAN(A1:J10;0.15)	473.95
=TRIMMEAN(A1:J10;0.2)	419.76
=TRIMMEAN(A1:J10;0.21)	419.76

من هذه المعادلات يتضح ما يلى:

- المعادله الاولى **=AVERAGE(A1:J10)** تقوم بحساب الوسط الحسابى للبيانات وكانت قيمته تساوى **699.83**.
- المعادله الثانيه **=TRIMMEAN(A1:J10;0)** تقوم بحساب الوسط المنسق لمجموعه البيانات مع استخدام نسبة حذف تساوى صفر، وهذا يعنى حساب الوسط الحسابى، ويلاحظ ان هذه المعادله تعطى نفس ناتج المعادله الاولى.
- المعادله الثالثه **=TRIMMEAN(A1:J10;0.05)** تقوم بحساب الوسط المنسق لمجموعه البيانات مع استخدام نسبة حذف تساوى **0.05** أى حذف خمس مفردات وحساب الوسط الحسابى لعدد 95 مفرده باقيه، وكانت قيمة الوسط المنسق تساوى **614.02**. لاحظ أن قيمة الوسط المنسق قد انخفضت قليلاً وهذا يعنى التخلص من بعض المفردات المتطرفه.
- المعادله الرابعه **=TRIMMEAN(A1:J10;0.1)** تقوم بحساب الوسط المنسق لمجموعه البيانات مع استخدام نسبة حذف تساوى **0.10** أى حذف عشر مفردات وحساب الوسط الحسابى لعدد 90 مفرده باقيه، وكانت قيمة الوسط المنسق تساوى **520.61**. لاحظ أن قيمة الوسط المنسق قد انخفضت قليلاً وهذا يعنى التخلص من بعض المفردات المتطرفه.

- المعادله الخامسه $=\text{TRIMMEAN}(A1:J10;0.15)$ تقوم بحساب الوسط المنسق لمجموعة البيانات مع استخدام نسبة حذف تساوى 0.15 أى حذف خمسة عشر مفردة وحساب الوسط الحسابى لعدد 85 مفردة باقيه، وكانت قيمة الوسط المنسق تساوى 473.95. لاحظ أن قيمة الوسط المنسق قد انخفضت قليلاً وهذا يعنى التخلص من مزيد من المفردات المتطرفه.
- المعادله السادسه $=\text{TRIMMEAN}(A1:J10;0.2)$ تقوم بحساب الوسط المنسق لمجموعة البيانات مع استخدام نسبة حذف تساوى 0.20 أى حذف عشرون مفردة وحساب الوسط الحسابى لعدد 80 مفردة باقيه، وكانت قيمة الوسط المنسق تساوى 419.76. لاحظ أن قيمة الوسط المنسق قد انخفضت قليلاً وهذا يعنى التخلص من مزيد من المفردات المتطرفه.
- المعادله السابعه $=\text{TRIMMEAN}(A1:J10;0.21)$ تقوم بحساب الوسط المنسق لمجموعة البيانات مع استخدام نسبة حذف تساوى 0.21 أى حذف واحد وعشرون مفردة وحساب الوسط الحسابى لعدد 79 مفردة باقيه، وكانت قيمة الوسط المنسق تساوى 419.76. لاحظ أن قيمة الوسط المنسق فى هذه الحاله تساوى الوسط المنسق فى حالة استخدام نسبة حذف تساوى 20% وهذا يعنى أن هناك 20% من البيانات تمثل قيم متطرفه ويجب ازالتها قبل عمل أى تحليل احصائى.

ويمكن للباحث تلخيص هذه النتائج فى الجدول التالى:

عدد المفردات	نسبة الحذف	عدد المفردات المحذوفه	الوسط المنسق
100	0%	صفر	699.83
100	5%	5	614.02
100	10%	10	520.61
100	15%	15	473.95
100	20%	20	419.76
100	21%	21	419.76

لاحظ أن الوسط المنسق عند استخدام نسبة حذف 20%، 21% متساويين، وهنا نحدد نسبة المفردات المتطرفه بالنسبه الاقل فيهما، وعليه فإن نسبة المفردات المتطرفه تساوى 20% من عدد المفردات الكلى.

(2) مقاييس التشتت

فى هذا الجزء سوف يتم التعرض لمقاييس التشتت باستخدام الدوال المتوافره بالبرنامج.

(1) المدى:

أن المدى هو الفرق بين اكبر قيمه واصغر قيمه، ونجد أن الاكسيل لا تتوفر فيه دالة المدى، ولكن يمكن استخدام دالة القيمه الكبرى Max، ودالة القيمه الصغرى Min، وايجاد الفرق بينهما لحساب المدى.

(1-1) دالة القيمه الكبرى:

هناك نوعان من الدوال لحساب القيمه الكبرى هذه الدوال هى:

◇ دالة القيمه الكبرى MAX لقيم رقميه فقط، تقوم بحساب القيمه الكبرى للخلايا الرقميه فقط وتتجاهل الخلايا الفارغه والخلايا النصيه.

◇ دالة القيمه الكبرى MAXA لقيم رقميه وقيم نصيه، هذه الداله تتعامل مع الارقام والنصوص، فاللفظ TRUE يعنى الرقم (1)، واللفظ FALSE يعنى الرقم (0)، واى لفظ اخر يعتبر صفر.

(11) مثال:

بفرض توافرت لدينا البيانات التاليه عن متغيرين:

	A	B	C	
1	X	Y		
2	18	0.2	=MAX(A2:A7)	18
3	12	0.3	=MAX(B2:B7)	0.9
4	13	0.5	=MAXA(B2:B10)	1
5	14	0.9		
6	9	0.1		
7	8	0.8		
8		TRUE		
9		FALSE		
10		Medhat		

يلاحظ من هذا الجدول ما يلى:

- المعادله الاولى =MAX(A2:A7) تقوم بحساب القيمه الكبرى لمجموعة خلايا رقميه.
- المعادله الثانيه =MAX(B2:B7) تقوم بحساب القيمه الكبرى للقيم الموجوده فى المدى B2:B7، ونجد أن القيمه الكبرى تساوى 0.9.

- المعادله الثالثه $=MAXA(B2:B10)$ تقوم بحساب القيمه الكبرى لمجموعه من القيم الرقميه والقيم النصيه، ويلاحظ أن اللفظ **True** يعنى واحد واللفظ **False** يعنى صفر، وبهذا نجد أن القيمه الكبرى لهذة الخلايا تساوى 1.